
Overview

 A new @-variable (@IDX.IOTYPE) has been added at 

UniVerse 11.1 which allows an index based on a 

BASIC subroutine to perform some functions similar 

to SQL based file triggers. 

 @IDX.IOTYPE can be accessed within an indexed 

subroutine to determine the type of I/O operation 

being performed. 

 An indexed subroutine will typically have less 

overhead than a SQL based trigger and may be more 

performant for certain operations.

UniVerse Indexed Subroutine Behavior

 What happens when writing/deleting a record in a file 

which has an Index based on a BASIC subroutine

 The subroutine is called once when adding a new record 

to the file or when deleting an existing record.

 On an update to an existing record, the subroutine is 

called twice. 

 First to evaluate the current value of the index.

 Second to evaluate the new value of the index.

 This is done to determine if an index update is needed.

UniVerse Indexed Subroutine Behavior

 Prior to 11.1, there was no way to determine what type 

of I/O operation was being performed (i.e. insert, 

update, or delete) while in the subroutine.

 At UniVerse release 11.1, the @variable @IDX.IOTYPE 

has been introduced.

 While the indexed subroutine is executing, the 

@IDX.IOTYPE  variable contains a numeric value 

corresponding to the type of I/O operation being 

performed. 

Possible Values for @IDX.IOTYPE

 0 is returned when checked outside an indexed 

subroutine

 1 is returned for an INSERT (i.e. when a new record is 

being added to the file)

 2 is returned for a DELETE (i.e. when an existing 

record is being deleted from the file)

Possible values for @IDX.IOTYPE

 3 is returned for an UPDATE when the subroutine is 

called to evaluate the original index value of an 

existing record (@RECORD contains original record     

contents)

 4 is returned for an UPDATE when the subroutine is 

called to evaluate the new index value of an existing 

record (@RECORD contains new record contents)

No Impact on Existing Applications

 This change does not alter how UniVerse indices 

functioned prior to 11.1. 

 The only change done at 11.1 related to this 

enhancement is that @IDX.IOTYPE is now available 

for use within an indexed BASIC subroutine.

 Indexes based on BASIC subroutines which do not 

use @IDX.IOTYPE will not be impacted.

Indexed Subroutine Example

0001       SUBROUTINE INDEX.SUB(RTNVAL)

0002       COMMON /INDEX.SUB/ OPENFLAG,F.AUDIT,OLDRECORD

0003       RTNVAL = "" ;* Set index value to "" for NO.NULLS index

0004       OPERATIONS = "INSERT":@FM:"DELETE":@FM:"UPDATE":@FM:"UPDATE"

0005       IF NOT(OPENFLAG) THEN

0006          OPEN "AUDIT.FILE" TO F.AUDIT ELSE STOP "CANNOT OPEN AUDIT.FILE"

0007          OPENFLAG = 1

0008       END

0009 *

(continued on next page)

0010 * The following case statement can be used to execute any specific

0011 * operations related to the type of operation being performed.

0012 *

0013       AUDIT.REC = ''

0014       BEGIN CASE

0015          CASE @IDX.IOTYPE = 1            ; * INSERT

0016          CASE @IDX.IOTYPE = 2            ; * DELETE

0017          CASE @IDX.IOTYPE = 3            ; * UPDATE BEFORE

0018             OLDRECORD = LOWER(@RECORD)

0019          CASE @IDX.IOTYPE = 4            ; * UPDATE AFTER

0020             AUDIT.REC<2> = OLDRECORD

0021          CASE 1

0022             RETURN

0023       END CASE

0024       IF @IDX.IOTYPE # 3 THEN

0025          RECID = @DATE:"*":SYSTEM(12):"*":@ID

0026          AUDIT.REC<1> = OPERATIONS<@IDX.IOTYPE>

0027          WRITE AUDIT.REC ON F.AUDIT,RECID

0028       END

0029       RETURN

0030    END

Creating an Index Subroutine

 BASIC  BP  INDEX.SUB

 CATALOG  BP  INDEX.SUB

 CT  DICT  TEST.IDX  INDEX.ITYPE

 0001: I

 0002: SUBR(INDEX.SUB)

 .

 CREATE.INDEX  TEST.IDX  INDEX.ITYPE  NO.NULLS

 BUILD.INDEX  TEST.IDX  INDEX.ITYPE

Performance Test Example

 Comparison testing done at 11.1.0 on AIX, HP, and 

11.1.1 on Windows platforms.

 Testing was done using an indexed or trigger 

subroutine which simply returned after being called. 

 Test program wrote 2 million records into both empty 

and full files using either index or trigger.

 Elapsed time to perform test was consistently 2 to 3 

times longer for trigger than indexed subroutine.

 Your mileage may vary.


